Seminar 19.05.2020

CONICE PE ECUATII GENERALE

Ex 1. Să se studieze şi să se reprezinte grafic următoarele conice pe ecuaţii generale, precizându-se (toate) elementele:

(a) $3x^2 + 2xy + 3y^2 + 8x - 8y + 8 = 0$

(b) $7x^2 - 8xy + y^2 - 6x - 12y - 9 = 0.$

In[1]:=
```
(*
    Folosesc El si nu E, fiindca in Mathematica litera E este protejata
*)
```

In[2]:=
```
(* polinomul de gradul al doilea care genereaza conica *)
```

In[3]:=
```
El[x_, y_] := 3 x^2 + 2 x y + 3 y^2 + 8 x - 8 y + 8;
```

Fie conica E(x,y)=0

In[4]:=
```
(* coeficientii formei patratice *)
```

In[5]:=
```
a11 = Coefficient[El[x, y], x^2];
a12 = Coefficient[El[x, y], x y] / 2;
a22 = Coefficient[El[x, y], y^2];
```

In[8]:=
```
(* matricea formei patratice *)
```

In[9]:=
```
A = {{a11, a12}, {a12, a22}};
```

In[10]:=
```
MatrixForm[A]
```

Out[10]//MatrixForm=

$$\begin{pmatrix} 3 & 1 \\ 1 & 3 \end{pmatrix}$$

In[11]:=
```
(* determinam genul conicei . . . *)
```

In[12]:=
```
δ = Det[A]
```

Out[12]= 8

In[13]:=
```
(* . . .  este pozitiv, deci conica este de gen eliptic *)
```

In[14]:=
```
(*
     Mai departe IA este I; in Mathematica simbolul I este protejat
*)
```

In[15]:=
```
IA = Tr[A]
```

Out[15]= 6

In[16]:=
```
a10 = Coefficient[Coefficient[El[x, y], x], y, 0] / 2;
a20 = Coefficient[Coefficient[El[x, y], y], x, 0] / 2;
```

In[18]:=
```
a00 = Coefficient[Coefficient[El[x, y], x, 0], y, 0];
```

In[19]:=
```
(*
   Simbolul D este protejat, foloses DE
*)
```

In[20]:=
```
DE = {{a11, a12, a10}, {a12, a22, a20}, {a10, a20, a00}}
```

Out[20]= {{3, 1, 4}, {1, 3, -4}, {4, -4, 8}}

In[21]:=
```
MatrixForm[DE]
```

Out[21]//MatrixForm=

$$\begin{pmatrix} 3 & 1 & 4 \\ 1 & 3 & -4 \\ 4 & -4 & 8 \end{pmatrix}$$

In[22]:=
```
Δ = Det[DE]
```

Out[22]= -64

diferit de 0, deci nedegenerata

In[23]:=
```
(* autovalorile si vectorii proprii matricii A *)
```

In[24]:=
```
λ = Eigenvalues[A];
```
det (A - λ I)=0

In[25]:=
```
λ1 = λ[[1]]
λ2 = λ[[2]]
```

Out[25]= 4

Out[26]= 2

In[27]:=
```
vect = Eigenvectors[A]
```
λ^2 - 6λ +8=0

Out[27]= {{1, 1}, {-1, 1}}

In[28]:=
```
nr1 = Sqrt[vect[[1]].vect[[1]]];
nr2 = Sqrt[vect[[2]].vect[[2]]];
```

In[30]:=
```
(* vectori proprii, normalizati *)
```

In[31]:=
```
v1 = vect[[1]] / nr1
v2 = vect[[2]] / nr2
```

Out[31]= $\left\{\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}$

Out[32]= $\left\{-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}\right\}$

In[33]:=
```
(* centrul elipsei *)
```

In[34]:=
```
eq1 = D[El[x, y], x]
eq2 = D[El[x, y], y]
```

Out[34]= 8 + 6 x + 2 y

Out[35]= -8 + 2 x + 6 y

In[36]:=
```
sol = Solve[{eq1 == 0, eq2 == 0}, {x, y}]
```

Out[36]= {{x → -2, y → 2}}

In[37]:=
```
(* VEZI SI SCHIMBAREA DE MAI JOS *)
```

In[38]:=
```
x0 = x /. sol[[1]];
y0 = y /. sol[[1]];
```
-2
2

In[40]:=
```
a = Sqrt[-Δ / (λ1 δ)]
b = Sqrt[-Δ / (λ2 δ)]
```

Out[40]= $\sqrt{2}$

Out[41]= 2

In[42]:=
```
c = If[a < b, Sqrt[b^2 - a^2], Sqrt[a^2 - b^2]]
```

Out[42]= $\sqrt{2}$

In[43]:=
```
(* axa focala *)
```

In[44]:=
```
v = If[a < b, v2, v1];
```

In[45]:=
```
r = 0.04;
```

In[46]:=
```
Centru = ParametricPlot[{x0, y0} + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Black, Thickness[0.01]}];
```

In[47]:=
```
V1 = ParametricPlot[{x0, y0} + a v1 + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Magenta, Thickness[0.005]}];
V2 = ParametricPlot[{x0, y0} + b v2 + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Orange, Thickness[0.005]}];
```

In[49]:=
```
V3 = ParametricPlot[{x0, y0} - a v1 + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Magenta, Thickness[0.005]}];
V4 = ParametricPlot[{x0, y0} - b v2 + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Orange, Thickness[0.005]}];
```

In[51]:=
```
(* focarele : se foloseste axa focala definita mai inainte *)
```

In[52]:=
```
F1 = ParametricPlot[{x0, y0} + c v + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Purple, Thickness[0.005]}];
F2 = ParametricPlot[{x0, y0} - c v + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Purple, Thickness[0.005]}];
```

In[54]:=
```
axa1 = ParametricPlot[{x0, y0} + t v1,
    {t, -a - 1, a + 1}, AspectRatio → Automatic, PlotStyle → Red];
axa2 = ParametricPlot[{x0, y0} + t v2, {t, -b - 1, b + 1},
    AspectRatio → Automatic, PlotStyle → Red];
```

In[56]:=
```
CX = Graphics[Arrow[{{x0, y0}, {x0, y0} + (a + .5) v1}]];
CY = Graphics[Arrow[{{x0, y0}, {x0, y0} + (b + 0.5) v2}]];
```

In[58]:=
```
axM = Graphics[{Red, Text["X", {x0, y0} + a v1 + {0.5, 0.75}]}];
axm = Graphics[{Red, Text["Y", {x0, y0} + b v2 + {-0.3, 0.5}]}];
```

In[60]:=
```
centrulC = Graphics[{Red, Text["C", {x0, y0} + {0, 0.2}]}];
```

In[61]:=
```
foc1 = Graphics[{Purple, Text["F", {x0, y0} + c v + {0, 0.2}]}];
foc2 = Graphics[{Purple, Text["F'", {x0, y0} - c v + {0, 0.2}]}];
```

In[63]:=
```
(* reprezentare grafica *)
```

In[64]:=
```
elipsa = ContourPlot[El[x, y] == 0, {x, -4, 1}, {y, -1, 4},
    Axes → True, AxesLabel → {x, y}, AspectRatio → Automatic];
```

In[65]:=
```
eqred = Graphics[{Brown,
    Text["Elipsa:", {x0 - 1, y0 - 2.5}], Text[ X^2/a^2 + Y^2/b^2 == 1, {x0, y0 - 2.5}]}];
```

In[66]:=
```
Show[elipsa, axa1, axa2, Centru, axM, axm, V1,
 V2, V3, V4, eqred, CX, CY, centrulC, F1, F2, foc1, foc2]
```

Out[66]=



In[67]:=
```
(*
   SCHIMBAREA COORDONATELOR, PENTRU A ADUCE CONICA LA FORMA REDUSA
*)
```

In[68]:=
```
(* ROTATIA : {x, y} --> {xp, yp} *)
```

In[102]:=
```
{x, y} = Transpose[{v1, v2}].{xp, yp}
```

Out[102]=
$$\left\{\frac{xp}{\sqrt{2}} - \frac{yp}{\sqrt{2}}, \frac{xp}{\sqrt{2}} + \frac{yp}{\sqrt{2}}\right\}$$

In[70]:=
```
(* ecuatia elipsei in noile coordonate xp, yp *)
```

In[103]:=
```
El1[xp_, yp_] := Simplify[El[x, y]] // Expand
```

In[104]:=
```
El1[xp, yp]
```

Out[104]=
$$8 + 4\,xp^2 - 8\sqrt{2}\,yp + 2\,yp^2$$

In[92]:=
```
(* A DOUA SCHIMBARE DE COORDONATE *)
```

In[117]:=
```
(* refacem coordonatele centrului in noile coordonate xp, yp *)
```

In[113]:=
```
{xp0, yp0} = {v1, v2}.{x0, y0}
```

Out[113]=
$$\left\{0, 2\sqrt{2}\right\}$$

In[118]:=
```
(* FACEM TRANSLATIA IN CENTRU *)
```

In[114]:=
```
xp = X + xp0
yp = Y + yp0
```

Out[114]=
$$X$$

Out[115]=
$$2\sqrt{2} + Y$$

In[116]:=
```
El2 = Simplify[El1[xp, yp]] // Expand
```

Out[116]=
$$-8 + 4\,X^2 + 2\,Y^2$$

In[1]:=
```
(* HIPERBOLA  *)

(* polinomul de gradul al doilea care genereaza hiperbola *)
```

In[2]:=
```
H[x_, y_] := 7 x^2 - 8 x y + y^2 - 6 x - 12 y - 9;
```

In[3]:=
```
(* coeficientii formei patratice *)
```

In[4]:=
```
a11 = Coefficient[H[x, y], x^2];
a12 = Coefficient[H[x, y], x y] / 2;
a22 = Coefficient[H[x, y], y^2];
```

In[7]:=
```
A = {{a11, a12}, {a12, a22}};
```

In[8]:=
```
MatrixForm[A]
```

Out[8]//MatrixForm=

$$\begin{pmatrix} 7 & -4 \\ -4 & 1 \end{pmatrix}$$

In[9]:=
```
(* avem gen hiperbolic *)
```

In[10]:=
```
δ = Det[A]
```

Out[10]=
```
-9
```

In[11]:=
```
a10 = Coefficient[Coefficient[H[x, y], x], y, 0] / 2;
a20 = Coefficient[Coefficient[H[x, y], y], x, 0] / 2;
```

In[13]:=
```
a00 = Coefficient[Coefficient[H[x, y], x, 0], y, 0];
```

In[14]:=
```
(*
   Simbolul D este protejat, foloses DE
*)
```

In[15]:=
```
DH = {{a11, a12, a10}, {a12, a22, a20}, {a10, a20, a00}}
```

Out[15]=
```
{{7, -4, -3}, {-4, 1, -6}, {-3, -6, -9}}
```

In[16]:=
```
MatrixForm[DH]
```

Out[16]//MatrixForm=

$$\begin{pmatrix} 7 & -4 & -3 \\ -4 & 1 & -6 \\ -3 & -6 & -9 \end{pmatrix}$$

In[17]:=
```
(* avem nedegenerare *)
```

In[18]:=
```
Δ = Det[DH]
```

Out[18]=
```
-324
```

In[19]:=
```
(* autovalorile si vectorii proprii matricii A *)
```

In[20]:=
```
λ = Eigenvalues[A];
```

In[21]:=
```
λ1 = λ[[1]]
λ2 = λ[[2]]
```

Out[21]=
```
9
```

Out[22]=
```
-1
```

In[23]:=
```
vect = Eigenvectors[A];
```

In[24]:=
```
nr1 = Sqrt[vect[[1]].vect[[1]]];
nr2 = Sqrt[vect[[2]].vect[[2]]];
```

In[26]:=
```
v1 = vect[[1]] / nr1
v2 = vect[[2]] / nr2
```

Out[26]= $\left\{-\frac{2}{\sqrt{5}}, \frac{1}{\sqrt{5}}\right\}$

Out[27]= $\left\{\frac{1}{\sqrt{5}}, \frac{2}{\sqrt{5}}\right\}$

In[28]:=
```
(* orientarea *)
```

In[29]:=
```
or = Det[{v1, v2}]
```

Out[29]= -1

In[30]:=
```
(*
    Daca orientarea eset negativa schimb semnul la v2
*)
```

In[31]:=
```
w2 = If[or > 0, v2, -v2];
v2 = w2
```

Out[32]= $\left\{-\frac{1}{\sqrt{5}}, -\frac{2}{\sqrt{5}}\right\}$

In[33]:=
```
(* centrul hiperbolei *)
```

In[41]:=
```
eq1 = D[H[x, y], x]
eq2 = D[H[x, y], y]
```

Out[41]= $-6 + 14\,x - 8\,y$

Out[42]= $-12 - 8\,x + 2\,y$

In[43]:=
```
sol = Solve[{eq1 == 0, eq2 == 0}, {x, y}]
```

Out[43]= $\{\{x \to -3, y \to -6\}\}$

In[44]:=
```
x0 = x /. sol[[1]];
y0 = y /. sol[[1]];
```

In[46]:=
```
c1 = -Δ / (λ1 δ)
c2 = -Δ / (λ2 δ)
```

Out[46]= -4

Out[47]= 36

In[48]:=
```
(* daca c1 > 0 axa focala este CX *)
```

In[49]:=
```
a = If[c1 > 0, Sqrt[-Δ / (λ1 δ)], Sqrt[Δ / (λ1 δ)]]
b = If[c1 > 0, Sqrt[Δ / (λ2 δ)], Sqrt[-Δ / (λ2 δ)]]
```

Out[49]= 2

Out[50]= 6

In[51]:=
```
c = Sqrt[a^2 + b^2];
```

In[52]:=
```
(*
        axa focala si d = a sau b
*)
```

In[53]:=
```
{v, d} = If[c1 > 0, {v1, a}, {v2, b}];
```

In[54]:=
```
r = 0.04;
```

In[55]:=
```
Centru = ParametricPlot[{x0, y0} + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Black, Thickness[0.01]}];
```

In[76]:=
```
V1 = ParametricPlot[{x0, y0} + d v + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Magenta, Thickness[0.005]}];
```

In[74]:=
```
V2 = ParametricPlot[{x0, y0} - d v + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Magenta, Thickness[0.005]}];
```

In[58]:=
```
(* focarele : se foloseste axa focala definita mai inainte *)
```

In[59]:=
```
F1 = ParametricPlot[{x0, y0} + c v + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Purple, Thickness[0.005]}];
F2 = ParametricPlot[{x0, y0} - c v + r {Cos[t], Sin[t]},
    {t, 0, 2 Pi}, PlotStyle → {Purple, Thickness[0.005]}];
```

In[61]:=
```
axa1 = ParametricPlot[{x0, y0} + t v1,
    {t, -a - 1, a + 1}, AspectRatio → Automatic, PlotStyle → Red];
axa2 = ParametricPlot[{x0, y0} + t v2, {t, -b - 1, b + 1},
    AspectRatio → Automatic, PlotStyle → Red];
```

In[63]:=
```
CX = Graphics[Arrow[{{x0, y0}, {x0, y0} + (a + 0.5) v1}]];
CY = Graphics[Arrow[{{x0, y0}, {x0, y0} + (b + 1.5) v2}]];
```

In[65]:=
```
axM = Graphics[{Red, Text["X", {x0, y0} + a v1 + {0.5, 0.75}]}];
axm = Graphics[{Red, Text["Y", {x0, y0} + b v2 + {-0.3, 0.5}]}];
```

In[67]:=
```
centrulC = Graphics[{Red, Text["C", {x0, y0} + {0, 0.2}]}];
```

In[68]:=
```
foc1 = Graphics[{Purple, Text["F", {x0, y0} + c v + {0, 0.2}]}];
foc2 = Graphics[{Purple, Text["F'", {x0, y0} - c v + {0, 0.2}]}];
```

In[70]:=
```
(* reprezentare grafica *)
```

In[71]:=
```
Hiperbola = ContourPlot[H[x, y] == 0, {x, -10, 3}, {y, -14, 4},
    Axes → True, AxesLabel → {x, y}, AspectRatio → Automatic];
```

In[72]:=
```
eqred = Graphics[{Brown, Text["Hiperbola:", {x0 - 5, y0 + 5}],
    Text[X^2/c1 + Y^2/c2 == 1, {x0 - 2, y0 + 5}]}];
```

In[77]:=
```
Show[Hiperbola, axa1, axa2, Centru, axM, axm,
 V1, V2, eqred, CX, CY, centrulC, F1, F2, foc1, foc2]
```

Out[77]=